

# The Shockley-Queisser limit

By Steve Byrnes. Last modified: 16 Oct 2012. This document lives at my website, <http://sjbyrnes.com/>. Please email me any feedback (my email address is on my website).

The Shockley-Queisser (SQ) limit is a famous limit on the maximal possible efficiency of solar cells, limited only by fundamental physics. It applies to most solar cell designs in the world, except for "tandem solar cells" and some additional obscure exceptions (discussed at the end of the document). The most important parameter in the SQ model is the bandgap of the semiconductor: If the gap is right, the efficiency can be up to 34%, if the gap is way off, the efficiency limit may be much smaller. The original SQ paper is <http://dx.doi.org/10.1063/1.1736034>, but it's also covered in every solar-cell textbook.

I'm using NREL's data for the solar spectrum (AM1.5G) and intensity (1000 W/m<sup>2</sup>). In the original SQ paper, they assumed that the sun had a 6000-kelvin blackbody spectrum. So my graphs and values are slightly different. However, other papers and books that use AM1.5G spectrum get the same results as I do, for example <http://www.opticsinfobase.org/abstract.cfm?URI=OSE-2010-SWA1>, <http://www.opticsinfobase.org/abstract.cfm?URI=OSE-2010-SWC4>, *Practical Handbook of Photovoltaics* p128-9, <http://dx.doi.org/10.1109/T-ED.1984.21594>.

I copied many of these graphs into the Wikipedia article on this topic ([http://en.wikipedia.org/wiki/Shockley-Queisser\\_limit](http://en.wikipedia.org/wiki/Shockley-Queisser_limit))

In this document you will find:

- \*A plot of the SQ efficiency limit as a function of bandgap
- \*A plot of the SQ limit on short-circuit current, on open-circuit voltage, and on fill-factor, as a function of bandgap
- \*A breakdown of exactly which factors lower the SQ limit for which bandgaps
- \*A list of some "loopholes" to exceed the SQ limit.

Enjoy!

Since someone asked me: "I release this document and code to the public domain."

Pronunciation of "Queisser": Hans-Joachim Queisser was German, so a German-speaker helped me guess how the name is pronounced. He guesses that "Queisser" rhymes with "nicer". ("Qu" as in "quick", "ei" as in "Einstein", "ss" as in "kiss", "er" as in "teacher"). (Thanks Florian!)

---

## General *Mathematica* program setup

Clear variables before starting:

```
In[1]:= ClearAll["Global`*"];
```

I wrote a *Mathematica* package that defines units and constants and allows unit-conversions. I'll be using it throughout. The unit and constant names should be self-explanatory. The package can be found at <http://sjbyrnes.com/>

```
In[2]:= << SteveUnits`
```

Turn off annoying error messages:

```
In[3]:= Off[NIntegrate::slwcon];
Off[NIntegrate::ncvb];
Off[FindMaximum::lstol];
```

## Program inputs

- **Solar cell temperature**

```
In[6]:= Tcell = 300 kelvin;
```

- Upload solar spectrum

Upload the NREL AM1.5G spectrum (Light from the whole sky at a typical latitude), from <http://rredc.nrel.gov/solar/spectra/am1.5/>. [Lots of data in a tiny font.]

```
[In]: AM15NoUnits=
```

Tack on units:

```
In[8]:= AM15 = ( {#[[1]] nm, #[[2]] watt/meter2/nm} & ) /@ AM15NoUnits;
```

The NREL data spans the following spectral range (in terms of both photon-wavelength and photon-frequency):


```
ln[9]: λMin = 280 nm;  
λMax = 4000 nm;  
Emin = hPlanck SpeedOfLight / λMax;  
Emax = hPlanck SpeedOfLight / λMin;
```

Interpolate to get a continuous function which I will be able to do integrals on:

```
In[13]:= AM15interp = Interpolation[AM15, InterpolationOrder -> 1];
```

Here's the plot, it looks correct:

```
In[14]:= Plot[AM15interp[λ nm] / (watt / meter2 / nm), {λ, 280, 4000}, PlotRange → All, AxesOrigin → {0, 0},
  Frame → True, FrameLabel → {"Wavelength (nm)", "Spectral intensity (W/m2/nm)"},
  PlotLabel → "Light from the sun"]
```



## Properties of incident sunlight

### ■ Put solar spectrum data in more convenient form

It's a bit more convenient for me to change the units for the solar spectrum, so that I can easily do integrals over photon energy, rather than wavelength, and calculate the number of photons instead of their energy. Therefore, I'll define the function "SPhotonsPerTEA" which stands for Solar Photons per unit Time, per unit photon Energy-range, per unit Area of the solar cell (assuming the cell is facing normal to the sun). To convert from the AM1.5 data to these new units, the formula is:

$$\text{SPhotonsPerTEA} \equiv \left( \frac{d(\text{number of photon per unit time per unit area})}{dE} \right) = \left( \frac{d(\text{photon power per unit area})}{d\lambda} \right) \times \frac{(\text{number of photons per unit time per unit area})}{(\text{photon power per unit area})} \times \left| \frac{d\lambda}{dE} \right| =$$

$$(\text{AM1.5 spectrum}) \times \frac{1}{\text{photon energy}} \times \frac{hc}{E^2}.$$

$$[\text{I used } \left| \frac{d\lambda}{dE} \right| = \left| \frac{d}{dE} \left( \frac{hc}{E} \right) \right| = \frac{hc}{E^2}].$$

```
In[15]:= SPhotonsPerTEA[Ephoton_] := AM15interp[ $\frac{h\text{Planck SpeedOfLight}}{\text{Ephoton}}$ ]  $\frac{1}{\text{Ephoton}}$   $\frac{h\text{Planck SpeedOfLight}}{\text{Ephoton}^2}$ 
```

Example: The following calculation means that there are  $1.43 \times 10^{18}$  solar photons with energy between 2eV and 2.001eV that hit a 1-square-meter patch each second:

```
In[16]:= SPhotonsPerTEA[2 eV] / (meter-2 meV-1 sec-1)
```

```
Out[16]= 1.42567 × 1018
```

Next: The "Solar constant" is the sun's total irradiance. If I did this right, it should be 1000 watts/meter<sup>2</sup>, because that's how NREL normalized their data.

```
In[17]:= SolarConstant = NIntegrate[Ephoton × SPhotonsPerTEA[Ephoton], {Ephoton, Emin, Emax}];
```

```
In[18]:= SolarConstant / (watt meter-2)
```

```
Out[18]= 1003.02
```

Close enough!

## ■ Photons above bandgap

For a given bandgap, I'm defining a function which is the total number of solar photons with energy above the bandgap, per unit time, per unit area on earth normal to the sun.

```
In[19]:= SolarPhotonsAboveGap[Egap_] := NIntegrate[SPhotonsPerTEA[Ephoton], {Ephoton, Egap, Emax}];
```

Here's an example: This means that  $2.76 \times 10^{21}$  photons with energy above 1.1 eV hit a one-square-meter patch of earth normal to the sun in one second.


```
In[20]:= SolarPhotonsAboveGap[1.1 eV] × 1 meter2 × 1 sec
```

```
Out[20]= 2.76024 × 1021
```

Here's a plot:

```
In[21]:= Temp1 = Table[{Egap / eV, SolarPhotonsAboveGap[Egap] / (cm-2 sec-1)}, {Egap, .4 eV, 3 eV, .05 eV}];
```

```
In[22]:= ListPlot[Temp1, ImageSize → 400, PlotRange → {0, 5 × 1017}, Joined → True,
  Frame → True, FrameLabel → {"Egap (eV)", "# Photons (cm-2 s-1)"},
  PlotLabel → "Number of above-bandgap photons as a function of bandgap"]
```




---

## Solar cell radiative recombinations and J-V curve

### ■ Recombination rate

In the best possible case, the only cause of electron-hole-pair recombination is radiative recombination. Radiative recombination occurs when an electron and hole collide, so it depends on how many electrons and holes there are, or more specifically it depends on the electron and hole QFLs.

## ■ Recombination rate when electron QFL = hole QFL ("QFL" is "Quasi-Fermi Level")

This is the case where electron QFL = hole QFL throughout the semiconductor. An example is the solar cell at zero bias in the dark. Then it's in thermal equilibrium and its radiation can be calculated by the blackbody formula -- more specifically, assuming it's a perfect blackbody above the bandgap and white-body below the bandgap. We also assume isotropic radiation from the top surface, and a mirror on the bottom surface.

Let RR0 be the "Radiative Recombination rate at 0 QFL splitting", (per solar-cell area). By the blackbody formula:

$$RR0 = \left( \frac{2\pi}{c^2 h^3} \int_{E_{gap}}^{\infty} \frac{E^2 dE}{\exp[E/k_B T_{cell}] - 1} \right)$$

In[23]:= `RR0[Egap_] :=`

$$\frac{2\pi}{\text{SpeedOfLight}^2 h \text{Planck}^3} \text{NIntegrate} \left[ \frac{E_{\text{photon}}^2}{\exp[E_{\text{photon}} / (k_B T_{\text{cell}})] - 1}, \{E_{\text{photon}}, E_{\text{gap}}, E_{\text{max}}\} \right]$$

## ■ Recombination rate when electron QFL and hole QFL are split

By kinetic theory, the radiative recombination rate is proportional to the product of electron concentration and hole concentration,  $p \times n$ . If you move the electron QFL up towards the conduction band by energy  $E$ , the electron concentration increases by  $\exp[-E/kT]$ . Likewise, if you move the hole QFL down towards the valence band by  $E$ , the hole concentration increases by  $\exp[E/kT]$ . Either way,  $p \times n \propto \exp[E/kT]$ , where  $E$  is the QFL energy splitting.

In the best possible case, the QFL splitting is equal to the external voltage (in reality, it may be larger than the external voltage). Therefore, the lowest possible radiative recombination rate is:

Recomb rate =  $e \times RR0 \exp[e V / k_B T_{cell}]$ , where  $V$  is the external voltage.

Note for pedants: I'm using the expression for radiative recombination  $\frac{2\pi}{c^2 h^3} \exp[e V / k_B T_{cell}] \int_{E_{gap}}^{\infty} \frac{E^2 dE}{\exp[E/k_B T_{cell}] - 1}$ . This isn't quite right: A more accurate expression is:  $\frac{2\pi}{c^2 h^3} \int_{E_{gap}}^{\infty} \frac{E^2 dE}{\exp[(E - e V)/k_B T_{cell}] - 1}$ . The difference is negligible except for tiny tiny bandgaps (less than 200meV). For explanation see <http://dx.doi.org/10.1109/T-ED.1980.19950> (Thanks Ze'ev!)

## ■ J-V curve

The current is from the electron-hole pairs that are created but which don't recombine. In the best case, all the solar photons possible are absorbed, while none recombine except radiatively. This gives:

$$J = e \times (\text{SolarPhotonsAboveGap} - RR0 (\exp[e V / k_B T_{cell}] - 1))$$

where  $J$  is the current per unit area, and  $V$  is the forward bias on the junction. The "-1" on the right accounts for spontaneous generation of e-h pairs through thermal fluctuations at 300K. I will leave out the "-1" below because  $RR0 \ll \text{SolarPhotonsAboveGap}$ .

$$\text{In[24]:= CurrentDensity}[V_, \text{Egap\_}] := e \left( \text{SolarPhotonsAboveGap}[\text{Egap}] - RR0[\text{Egap}] \exp\left[\frac{e V}{k_B T_{cell}}\right] \right)$$

## ■ Open-circuit voltage, short-circuit current density

In[25]:= `Jsc[Egap_] := CurrentDensity[0, Egap];`

$$\text{Voc}[Egap_] := \frac{k_B T_{cell}}{e} \log \left[ \frac{\text{SolarPhotonsAboveGap}[Egap]}{RR0[Egap]} \right];$$

Example: An ideal 1.1eV-bandgap solar cell has a short-circuit current of 44 mA/cm<sup>2</sup>, and an open-circuit voltage of 0.86V.


In[27]:= `{Jsc[1.1 eV] / (mA cm^-2), Voc[1.1 eV] / volt}`

Out[27]= {44.2239, 0.857749}

Plot:


```
In[28]:= Temp2 = Table[{Egap / eV, Jsc[Egap] / (mA / cm2)}, {Egap, .4 eV, 3 eV, .05 eV}];
```

```
In[29]:= ListPlot[Temp2, ImageSize → 400, PlotRange → {0, 70},
  Joined → True, Frame → True, FrameLabel → {"Egap (eV)", "Jsc (mA/cm2)"},
  PlotLabel → "Ideal short-circuit current as a function of bandgap"]
```



```
In[30]:= Temp3 = Table[{Egap / eV, Voc[Egap] / volt}, {Egap, .4 eV, 3 eV, .05 eV}];
```

```
In[31]:= Show[ListPlot[Temp3, ImageSize → 400, PlotRange → {0, 3},
  Joined → True, Frame → True, FrameLabel → {"Egap (eV)", "Voc (V)"}, PlotLabel →
  "Ideal open-circuit voltage as a function of bandgap (Dashed line is bandgap)"],
  Plot[x, {x, 0.4, 3}, PlotStyle → {Dashed, Purple}]]
```



## Ideal bandgap and maximum efficiency

Given what we've already done, it's now simple to calculate the ideal bandgap and efficiency, by numerically maximizing the product  $IV$  for each bandgap. The efficiency is the product of current density and voltage, divided by the solar constant (i.e. incoming light power)

### ■ Maximum power point (MPP)

```
In[32]:= MaxPower[Egap_] := FindMaximum[(V × CurrentDensity[V, Egap]), {V, 0}][[1]];
VatMPP[Egap_] := V /. FindMaximum[(V × CurrentDensity[V, Egap]), {V, 0}][[2]];
JAtMPP[Egap_] := CurrentDensity[VatMPP[Egap], Egap];
MaxEfficiency[Egap_] := MaxPower[Egap] / SolarConstant;
```

Example:


```
In[36]:= MaxEfficiency[1.1 eV]
```

```
Out[36]= 0.32811
```

Plot: The famous SQ efficiency limit!

```
In[37]:= Temp4 = Table[{Egap / eV, MaxEfficiency[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}];
```

```
In[38]:= ListPlot[Temp4, ImageSize → 400, PlotRange → {0, .35},
  Joined → True, Frame → True, FrameLabel → {"Egap (eV)", "Max efficiency"},
  PlotLabel → "SQ efficiency limit as a function of bandgap"]
```




### ■ Fill factor

```
In[39]:= FF[Egap_] :=  $\frac{\text{MaxPower}[Egap]}{\text{Jsc}[Egap] \text{ Voc}[Egap]}$ ;
```

Plot:

```
In[40]:= Temp5 = Table[{Egap / eV, FF[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}];
```

```
In[41]:= ListPlot[Temp5, ImageSize → 400, PlotRange → {0, 1},
  Joined → True, Frame → True, FrameLabel → {"Egap (eV)", "Fill factor"},
  PlotLabel → "Ideal fill factor as a function of bandgap"]
```



## Quantifying losses

### ■ Quantifying losses

We split the incoming light power into five parts:

- (A) Power converted into useful electricity;
- (B) Power of below-bandgap photons, which is wasted power because these photons are not absorbed;
- (C) Excess photon energy beyond the bandgap, which is wasted because the electron and hole just immediately relax to the band edges. For example, for a 1eV-bandgap semiconductor, a 3eV photon creates the same electron-hole pair as a 1.01eV photon. All the 2eV of extra energy carried by a 3eV photon in that case is wasted.
- (D) Power lost due to electron-hole recombination at the max-power-point, which is wasted as heat;
- (E) Power lost because the voltage of the cell at the max-power-point is less than the bandgap.

To say the same thing using equations:

Light power in =

$$\begin{aligned}
 & (V_{\text{MPP}} \times I_{\text{MPP}}) \\
 & + (\text{Power of below-bandgap photons}) \\
 & + (\text{Power of above-bandgap photons} - \text{Number of above-bandgap photons} \times \text{Bandgap energy}) \\
 & + ((\text{Number of above-bandgap photons} - I_{\text{MPP}}/e) \times (\text{Bandgap energy})) \\
 & + I_{\text{MPP}} \times (\text{Bandgap voltage} - V_{\text{MPP}})
 \end{aligned}$$

I'll write everything as a fraction of the incident light power.



```

In[42]:= UsefulElectricity[Egap_] := MaxEfficiency[Egap];
BelowGap[Egap_] := 
$$\frac{N \text{Integrate}[E_{\text{photon}} \times S_{\text{PhotonsPerTEA}}[E_{\text{photon}}], \{E_{\text{photon}}, E_{\text{min}}, E_{\text{gap}}\}]}{\text{SolarConstant}};$$

ExcessBeyondGap[Egap_] := 
$$\frac{N \text{Integrate}[(E_{\text{photon}} - E_{\text{gap}}) \times S_{\text{PhotonsPerTEA}}[E_{\text{photon}}], \{E_{\text{photon}}, E_{\text{gap}}, E_{\text{max}}\}]}{\text{SolarConstant}};$$

MPPRecombination[Egap_] := 
$$\frac{(\text{SolarPhotonsAboveGap}[E_{\text{gap}}] - J_{\text{AtMPP}}[E_{\text{gap}}] / e) \times E_{\text{gap}}}{\text{SolarConstant}};$$

MPPVoltageIsLessThanGap[Egap_] := 
$$\frac{J_{\text{AtMPP}}[E_{\text{gap}}] \times (E_{\text{gap}} / e - V_{\text{AtMPP}}[E_{\text{gap}}])}{\text{SolarConstant}};$$


```

Example: Accounting for the energy and losses of an ideal 1.1eV solar cell. Everything adds up to 100% (within numerical accuracy) as expected.

```

In[47]:= {UsefulElectricity[1.1 eV], BelowGap[1.1 eV], ExcessBeyondGap[1.1 eV],
MPPRecombination[1.1 eV], MPPVoltageIsLessThanGap[1.1 eV]}

```

```

Out[47]= {0.32811, 0.187294, 0.325116, 0.0157707, 0.141118}

```

```

In[48]:= Total[%]

```

```

Out[48]= 0.997408

```

Plot:

```

In[49]:= Temp6 =
{Table[{Egap / eV, UsefulElectricity[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}],
Table[{Egap / eV, BelowGap[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}],
Table[{Egap / eV, ExcessBeyondGap[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}],
Table[{Egap / eV, MPPRecombination[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}],
Table[{Egap / eV, MPPVoltageIsLessThanGap[Egap]}, {Egap, .4 eV, 3 eV, .05 eV}]};

```

The first command below calculates partial sums of Temp6, so the plots will stack on top of each other. (Sorry that the code is incomprehensible.)

```

In[50]:= Temp6partialsums = Transpose[
  Table[Rest[FoldList[{#2[[1]], #1[[2]] + #2[[2]]} &, {0, 0}, (Transpose[Temp6][[i]])]],
    {i, 1, Length[Temp6[[1]]]}];
ListPlot[Temp6partialsums, Joined → True, Filling →
  {1 → {0, Black}, 1 → {{2}, Pink}, 2 → {{3}, Green}, 3 → {{4}, Blue}, 4 → {{5}, Gray}},
  Frame → True, FrameLabel → {"Bandgap (eV)", "Fraction of incoming light power"},
  PlotLabel → "...POWER GOES TO...\nUseful electricity (black);\nBelow-gap
    photons (pink);\ne-h relaxation to the band edges (green);\nCurrent loss
    from radiative recombination (blue)\nVoltage is less than bandgap (gray)"]

```

...POWER GOES TO...


Useful electricity (black);

Below-gap photons (pink);

e-h relaxation to the band edges (green);

Current loss from radiative recombination (blue)

Voltage is less than bandgap (gray)



Plot the same thing using pie charts instead:

```

In[52]:= Needs["PieCharts`"]

```


```

In[53]:= ShowPieChartLosses[Egap_] := PieChart[
  {UsefulElectricity[Egap], BelowGap[Egap], ExcessBeyondGap[Egap], MPPRecombination[Egap],
    MPPVoltageIsLessThanGap[Egap]}, PieLabels → {"Useful electricity", "Photons below gap",
    "Photon excess energy\nbeyond gap", "e-h recombination", "VMPP < Vgap"},
  PlotLabel → "LOSSES FOR BANDGAP " <> ToString[Egap/eV] <> "eV", ImageSize → 200]


```

```
In[54]:= {ShowPieChartLosses[0.5 eV], ShowPieChartLosses[1 eV],
ShowPieChartLosses[1.5 eV], ShowPieChartLosses[2 eV], ShowPieChartLosses[2.5 eV]}
```


LOSSES FOR BANDGAP 0.5eV




LOSSES FOR BANDGAP 1.eV




LOSSES FOR BANDGAP 1.5eV



LOSSES FOR BANDGAP 2.eV



LOSSES FOR BANDGAP 2.5eV



## Partial list of "loopholes" to exceed the SQ limit

Exceeding the SQ limit is a sort of obsession of solar cell physicists. The standard reference book on this subject is *Third Generation Photovoltaics* by Martin Green. Here is a partial list.

### ■ Tandem solar cells

Tandem solar cells can and do exceed the SQ limit. In the SQ model, it is assumed that photons with energy below the bandgap are not absorbed at all, and with energy far above the bandgap are absorbed but all that excess energy is wasted. A tandem solar cell allows high-energy photons to excite a large-bandgap solar cell, and lower-energy photons to excite a smaller-bandgap solar cell. Usually 2 or 3 or 4 solar cells are used. Tandem solar cells are widely used and available. They are used commercially to make the very highest-efficiency most-expensive solar cells. They are also used commercially in some lower-end solar cells, such as amorphous silicon.

In the above derivation, where exactly is the loophole that tandem cells take advantage of? It is the step where I wrote "In the best possible case, the QFL splitting is equal to the external voltage (in reality, it may be larger than the external voltage)." Normally, this is true because, to get a net flow of photogenerated electrons into the cathode lead, the electron QFL has to tilt downward towards the cathode lead, and likewise to get a net flow of photogenerated holes into the anode lead, the hole QFL has to tilt upwards towards the cathode lead. However, in a tandem cell, the QFLs repeatedly split apart and come back together, such that the external voltage can be much greater than the QFL splitting at any given point.

### ■ Light concentration

The sun is a small point in the sky, so sunlight can be focused to very high concentrations (in theory as much as 50,000X!) Higher concentrations generally lead to higher efficiencies, assuming the solar panel does not heat up too much. In other words, if you focus sunlight to make it 100X more intense on a solar cell, the power can go up by *more* than 100X in the SQ model. Although light concentration is commonly used in practice for solar cells, the possible modest efficiency gain is *not* the *primary* reason that concentrated solar cells are used, in my understanding. More important is that small solar cells with huge mirrors/lenses can sometimes be cheaper than huge solar cells with no mirrors/lenses.

### ■ Anisotropic radiation

If you engineer a solar cell to only radiate in one direction, its theoretical efficiency increases. (The direction it radiates has to point towards the sun, otherwise light cannot get in.) The theoretical increase here is the same as if you were doing light-concentration. Unfortunately, in practice, it is not as helpful as you might hope, because it reduces radiative recombination but does not reduce nonradiative recombination.

### ■ Hot electrons

Electrons are usually excited to an energy far above the conduction-band-minimum, and holes far below the valence-band-maximum. Usually they relax to the band edges very quickly. If you can stop that relaxation, you can theoretically exceed the SQ limit by a lot. It seems to be very difficult in practice, and so far there is not even a proof-of-principle laboratory demonstration of a complete hot-electron device.

## ■ Multiple-exciton generation

In the SQ calculation, it is assumed that each photon above the bandgap creates just one electron-hole pair. In theory, a photon at twice the bandgap could produce two e-h pairs, at 3X the bandgap could produce 3, etc. This is called "Multiple-exciton generation" (MEG). There has been a lot of work on this in recent years, particularly in quantum dots, and for many years there was controversy over whether MEG had been observed. The "smoking gun" for MEG---i.e., unarguable proof of its occurrence---is a device with "external quantum efficiency" above 100% (i.e., more than one electron flows through the ammeter, for each photon coming into the device). After a lot of work, that milestone was finally reached in December 2011: <http://dx.doi.org/10.1126/science.1209845>. So now we know for sure that MEG is possible in some devices. Of course, we are still a long way away from seeing a commercial device that takes advantage of MEG.

## ■ Intermediate-band solar cell

In the SQ calculation, it is assumed that photons below the bandgap are wasted. Instead, it is possible to have an energy level in the gap, and excite electrons from the valence to conduction band in two steps. There have been a few laboratory proof-of-principle demonstrations of the concept: <http://dx.doi.org/10.1103/PhysRevLett.106.028701>, <http://dx.doi.org/10.1063/1.3166863>, <http://dx.doi.org/10.1103/PhysRevLett.97.247701>. So far there is no demonstration that intermediate-band solar cells can have high efficiencies in practice.

## ■ Chemical upconversion

In the SQ calculation, it is assumed that photons below the bandgap are wasted. Instead, it is possible to incorporate an upconversion chemical that absorbs two low-energy photons and emits one high-energy photon. Such chemicals exist, and upconverting solar cells have been demonstrated, but so far these have not been incorporated into commercial solar cells, I think primarily because the chemicals are expensive rare-earth compounds, and they can only upconvert within a narrow wavelength range. (But I'm not sure.)

## ■ Chemical downconversion

In the SQ calculation, it is assumed that each photon above the bandgap creates just one electron-hole pair. In theory, a photon at twice the bandgap could produce two e-h pairs, at 3X the bandgap could produce 3, etc. The idea of "downconversion" is to incorporate a downconversion chemical that absorbs one very-high-energy photon and re-radiates two lower-energy photons (lower-energy but still above the bandgap) that add up to the energy of the original photon. The system is only worthwhile if each incoming high-energy photon is converted to *more than 1 on average* outgoing above-bandgap photon. This requires high downconversion efficiency and very low loss, and this benchmark that has not yet been reached (as far as I know).

## ■ Non-blackbody solar cell and "photon recycling"

Say you have a material with too small a bandgap, like 0.5eV, but you want it to use it in a solar cell anyway. You are at a disadvantage because the Shockley-Queisser efficiency limit for 0.5eV bandgap is much lower than the limit for, say, 1.2eV bandgap. However, what you can do is put your solar cell behind a dichroic mirror or coating which reflects all light with frequency below 1.2eV and transmits all light with frequency above 1.2eV. If you do this, the 0.5eV-bandgap material behaves *as if* it had a 1.2eV bandgap, and in particular it has the same theoretical efficiency limit as if it had a 1.2eV bandgap. What's happening is that most of the photons emitted in electron-hole recombination events in the semiconductor are reflected right back into the semiconductor, where they are reabsorbed into a new electron-hole pair. Therefore you can get a huge density of electrons and holes with very little *net* radiative recombination. These electrons and holes fill up the conduction and valence bands until the "Dynamic Burstein-Moss"-shifted bandgap is close to 1.2eV. Another way to look at it is, you have made the material into a whitebody below 1.2eV and a blackbody above 1.2eV, which is the same as a 1.2eV bandgap material. See <http://dx.doi.org/10.1063/1.3682101> for more details.

More generally, when a radiative recombination photon is re-absorbed to make a new electron-hole pair, as in the previous paragraph, it's sometimes called "Photon recycling". Photon recycling is important for modeling solar cells, for example [http://dx.doi.org/10.1016/0927-0248\(93\)90142-P](http://dx.doi.org/10.1016/0927-0248(93)90142-P). However, photon recycling cannot on its own make a solar cell overcome the overall Shockley-Queisser limit of ~34%. (An exception is the anisotropic radiation design, see above.) Moreover, photon recycling reduces net radiative recombination, but does not reduce nonradiative recombination, so it's not as helpful as you might otherwise hope.